

REPRESENTATIONS FOR PARTICLES UP TO SPIN $3/2$

THE GOSAM COLLABORATION
THOMAS REITER

ABSTRACT. This document describes the implementation of wave-functions and propagators in **GOLEM**.

CONTENTS

1. Global Structure	1
2. Spin-0 Particles	4
3. Spin- $1/2$ Particles	4
3.1. Massless Case	5
3.2. Massive Case	6
4. Spin-1 Particles	8
4.1. Massless Case	8
4.2. Massive Case	10
5. Spin- $\frac{3}{2}$ Particles	12
6. Spin-2 Particles	17
6.1. Tensor Structure	17
6.2. Tensor Ghost	17
6.3. Gravitons	18
7. The Colour Part of the propagators	21
Appendix A. Index of Symbols	21
Appendix B. Index of Macros	21
Appendix C. Index of Files	22
References	22

1. GLOBAL STRUCTURE

The replacements for the wave-functions go into the file **legs.hh**, propagators are found in **propagators.hh**. The diagram generator is expected to yield the following functions:

inplorentz($2s, i, k, m$): for each initial state particle of spin s , momentum k and mass m . The index i is a Lorentz index in the corresponding representation which connects the wave function to the rest of the diagram. In cases where particle and

antiparticle are distinct, the parameter $2s$ is signed ($-2s$ for the antiparticle).

`outlorentz($2s, i, k, m$)`: as above, but for each final state particle.

`proplorentz($2s, k, m, \Gamma, A, i_1, i_2$)`: denotes the Lorentz part of a propagator for a particle of spin s , momentum k and mass m . The indices i_1 and i_2 are corresponding Lorentz indices. The decay width of the particle is Γ . The parameter A is a flag that indicates special properties of a field and is non-zero if the propagator needs special treatment.

`inpcolor(n, i)`: for each initial state particle. Associates the colour index i with the initial state particle number n . This function is not treated in this file.

`outcolor(n, i)`: for each final state particle. Associates the colour index i with the final state particle number n . This function is not treated in this file.

`propcolor(r, i_1, i_2)`: denotes the colour part of a propagator, where r is a representation label; r is either 1 (trivial rep.), -3 or 3 (fundamental rep.) or 8 (adjoint rep.). The indices i_1 and i_2 are the colour indices of that propagator.

`inp($f, k, [h], [k^b], [q]$)`: carries the helicity information h of an initial state particle of the field f and momentum k . For massive gauge bosons, the parameters k^b and q are the two momenta of the light-cone splitting. For massless gauge the parameter k^b is omitted. The parameters h, k^b and q are not generated by the diagram generator but added at an earlier point in the `Form` program.

`out($f, k, [h], [k^b], [q]$)`: same as `inp` but for final state particles.

On the output side we use the symbols introduced by the `spinney` library plus the scalar propagator

(1)

$$\text{inv}(k, m) = \frac{1}{k^2 - m^2 + i0^+} \quad \text{and} \quad \text{inv}(k, m, \Gamma) = \frac{1}{k^2 - m^2 - im\Gamma + i0^+}$$

`< common header 2a >` \equiv

```
* vim: ts=3:sw=3
* This file is generated from lorentz.nw.
* Do not edit this file directly.
* Instead change src/form/lorentz.nw and run make.
```

◇

Macro referenced in [2b](#), [3](#).

`"legs.hh" 2b` \equiv

```
< common header 2a >
*---#[ Scalars :
```

```

⟨ scalar wave-functions 4a ⟩
*---#] Scalars :
*---#[ Spinors :
*---#[   Massless Spinors :
⟨ wave-functions for massless spinors 5c ⟩
*---#]   Massless Spinors :
*---#[   Massive Spinors :
⟨ wave-functions for massive spinors 7a ⟩
*---#]   Massive Spinors :
*---#] Spinors :
*---#[ Polarisation Vectors for Gauge Bosons :
*---#[   Massless Gauge Bosons :
⟨ gauge boson wave-functions, light-like 9b ⟩
*---#]   Massless Gauge Bosons :
*---#[   Massive Gauge Bosons :
⟨ gauge boson wave-functions, massive 11d ⟩
*---#]   Massive Gauge Bosons :
*---#] Polarisation Vectors for Gauge Bosons :
*---#[ wave functions for Vector-Spinors :
Repeat;
  ⟨ vector-spinor wave functions 13 ⟩
EndRepeat;
*---#] wave functions for Vector-Spinors :
*---#[ wave functions for gravitons :
Repeat;
  ⟨ graviton wave functions 19b ⟩
EndRepeat;
*---#] wave functions for gravitons :
◇

```

"propagators.hh" 3≡

```

  ⟨ common header 2a ⟩
  ⟨ colour part of the propagators 21 ⟩
  *---#[ Scalar Bosons :
  ⟨ scalar propagator 4b ⟩
  *---#[ Scalar Bosons :
  *---#[ Fermions :
  ⟨ fermion propagator 5a ⟩
  ⟨ handed fermion propagator 5b ⟩
  *---#[ Fermions :
  *---#[ Gauge Bosons :
  ⟨ gauge boson propagator 8a ⟩
  *---#[ Gauge Bosons :
  *---#[ Vector-Spinor propagator :
  ⟨ vector-spinor propagators 12 ⟩
  *---#[ Vector-Spinor propagator :
  *---#[ Tensor Bosons :

```

```

Repeat;
  < tensor ghost propagator 18a >
  < graviton propagator 18b >
  < graviton effective propagator 19a >
EndRepeat;
*---#] Tensor Bosons :
◇

```

For the Feynman rules we stick to the conventions of [1].

2. SPIN-0 PARTICLES

The wave function of a spin-0 particle is represented by a pure number.

```

< scalar wave-functions 4a > ≡
  ld inplorentz(0, iv?, k1?, m?) = 1;
  ld outlorentz(0, iv?, k1?, m?) = 1;◇
Macro referenced in 2b.

```

Its propagator is just the plain propagator

$$(2) \quad \frac{i}{k^2 - m^2 - im\Gamma + i0^+}.$$

```

< scalar propagator 4b > ≡
  ld proplorentz(0, k1?, m?, sDUMMY1?, 0, iv1?, iv2?) =
    PREFACTOR(i_) * inv(k1, m, sDUMMY1);
  ld proplorentz(0, 0, m?, sDUMMY1?, 0, iv1?, iv2?) =
    PREFACTOR(i_) * inv(ZERO, m, sDUMMY1);◇
Macro referenced in 3.

```

3. SPIN-1/2 PARTICLES

For spinor wave functions we have the following assignment in the notation of [1]:

	l^-, q	l^+, \bar{q}
initial	$u_\alpha(k, j_3)$	$\bar{v}_\alpha(k, j_3)$
final	$\bar{u}_\alpha(k, j_3)$	$v_\alpha(k, j_3)$

Here, l and q stand for leptons and quarks respectively. The index α denotes a spinor index and j_3 is the 3-component of the spin. We label the states by $j_3 = \pm 1$ instead of the physical values $j_3 = \pm 1/2$.

The propagator both for the massive and the massless case is

$$(3) \quad \frac{i(\not{k} + m)_{\alpha\beta}}{k^2 - m^2 - im\Gamma + i0^+}$$

where the momentum flow is from β to α .

$\langle \text{fermion propagator 5a} \rangle \equiv$
`ld proplorentz(1, k1?, m?, sDUMMY1?, 0, iv1?, iv2?) =`
`PREFACTOR(i_) * (NCContainer(Sm(k1), iv2, iv1)`
`+ csqrt(m*(m-i_*sDUMMY1)) * NCContainer(1, iv2, iv1)`
`) * inv(k1, m, sDUMMY1);`
`ld proplorentz(1, 0, m?, sDUMMY1?, 0, iv1?, iv2?) =`
`+ PREFACTOR(i_ * csqrt(m*(m-i_*sDUMMY1))) * NCContainer(1, iv2, iv1) * in`
 Macro referenced in [3](#).

For massless fermions, the auxilliary field can also have values 1 and -1 for left- and right-handed particles. This follows the CalcHEP convention.

Golem	CalcHEP	Expression
+1	'L'	$\frac{\not{p}\Pi_+}{p^2}$
-1	'R'	$\frac{\not{p}\Pi_-}{p^2}$

$\langle \text{handed fermion propagator 5b} \rangle \equiv$
`ld proplorentz(1, k1?, 0, 0, 1, iv1?, iv2?) =`
`PREFACTOR(i_) * NCContainer(Sm(k1)*ProjPlus, iv2, iv1) * inv(k1, 0);`
`ld proplorentz(1, k1?, 0, 0, -1, iv1?, iv2?) =`
`PREFACTOR(i_) * NCContainer(Sm(k1)*ProjMinus, iv2, iv1) * inv(k1, 0);◇`
 Macro referenced in [3](#).

3.1. Massless Case. For massless spinors we translate the spin states directly into helicity eigenstates as follows¹:

$$\begin{aligned}
 (4a) \quad & u_\alpha(k, +1) = |k\rangle & \bar{u}_\alpha(k, +1) = [k| \\
 (4b) \quad & u_\alpha(k, -1) = |k\rangle & \bar{u}_\alpha(k, -1) = \langle k| \\
 (4c) \quad & v_\alpha(k, +1) = |k\rangle & \bar{v}_\alpha(k, +1) = \langle k| \\
 (4d) \quad & v_\alpha(k, -1) = |k\rangle & \bar{v}_\alpha(k, -1) = [k|
 \end{aligned}$$

$\langle \text{wave-functions for massless spinors 5c} \rangle \equiv$
 $\langle \text{implementation of Equation (4a) 6a} \rangle$
 $\langle \text{implementation of Equation (4b) 6b} \rangle$
 $\langle \text{implementation of Equation (4c) 6c} \rangle$
 $\langle \text{implementation of Equation (4d) 6d} \rangle \diamond$
 Macro referenced in [2b](#).

¹Please, refer to the **spinney** documentation for notational conventions of bra- and ket-spinors.

$\langle \text{implementation of Equation (4a) 6a} \rangle \equiv$
`ld inplorentz(1, iv?, k1?, 0) *
inp(field1?, k1?, 1) =
NCContainer(USpa(k1), iv);
ld outlorentz(1, iv?, k1?, 0) *
out(field1?, k1?, 1) =
NCContainer(UbarSpb(k1), iv);`◇
Macro referenced in 5c.

$\langle \text{implementation of Equation (4b) 6b} \rangle \equiv$
`ld inplorentz(1, iv?, k1?, 0) *
inp(field1?, k1?, -1) =
NCContainer(USpb(k1), iv);
ld outlorentz(1, iv?, k1?, 0) *
out(field1?, k1?, -1) =
NCContainer(UbarSpa(k1), iv);`◇
Macro referenced in 5c.

$\langle \text{implementation of Equation (4c) 6c} \rangle \equiv$
`ld outlorentz(-1, iv?, k1?, 0) *
out(field1?, k1?, 1) =
NCContainer(USpb(k1), iv);
ld inplorentz(-1, iv?, k1?, 0) *
inp(field1?, k1?, 1) =
NCContainer(UbarSpa(k1), iv);`◇
Macro referenced in 5c.

$\langle \text{implementation of Equation (4d) 6d} \rangle \equiv$
`ld outlorentz(-1, iv?, k1?, 0) *
out(field1?, k1?, -1) =
NCContainer(USpa(k1), iv);
ld inplorentz(-1, iv?, k1?, 0) *
inp(field1?, k1?, -1) =
NCContainer(UbarSpb(k1), iv);`◇
Macro referenced in 5c.

3.2. Massive Case. Massive spinors translate to spinney notation in the following sense:

$$\begin{aligned}
(5a) \quad u_\alpha(k, +1) &= |k^+\rangle & \bar{u}_\alpha(k, +1) &= [k^+| \\
(5b) \quad u_\alpha(k, -1) &= |k^+\rangle & \bar{u}_\alpha(k, -1) &= \langle k^+| \\
(5c) \quad v_\alpha(k, +1) &= |k^-\rangle & \bar{v}_\alpha(k, +1) &= \langle k^-| \\
(5d) \quad v_\alpha(k, -1) &= |k^-\rangle & \bar{v}_\alpha(k, -1) &= [k^-|
\end{aligned}$$

$\langle \text{wave-functions for massive spinors } 7a \rangle \equiv$
 $\langle \text{implementation of Equation (5a) } 7b \rangle$
 $\langle \text{implementation of Equation (5b) } 7c \rangle$
 $\langle \text{implementation of Equation (5c) } 7d \rangle$
 $\langle \text{implementation of Equation (5d) } 7e \rangle \diamond$
 Macro referenced in 2b.

$\langle \text{implementation of Equation (5a) } 7b \rangle \equiv$
ld inplorentz(1, iv?, k1?, m?) *
 inp(field1?, k1?, 1) =
 NCContainer(USpa(k1, +1), iv);
ld outlorentz(1, iv?, k1?, m?) *
 out(field1?, k1?, 1) =
 NCContainer(UbarSpb(k1, +1), iv); \diamond
 Macro referenced in 7a.

$\langle \text{implementation of Equation (5b) } 7c \rangle \equiv$
ld inplorentz(1, iv?, k1?, m?) *
 inp(field1?, k1?, -1) =
 NCContainer(USpb(k1, +1), iv);
ld outlorentz(1, iv?, k1?, m?) *
 out(field1?, k1?, -1) =
 NCContainer(UbarSpa(k1, +1), iv); \diamond
 Macro referenced in 7a.

$\langle \text{implementation of Equation (5c) } 7d \rangle \equiv$
ld outlorentz(-1, iv?, k1?, m?) *
 out(field1?, k1?, 1) =
 NCContainer(USpb(k1, -1), iv);
ld inplorentz(-1, iv?, k1?, m?) *
 inp(field1?, k1?, 1) =
 NCContainer(UbarSpa(k1, -1), iv); \diamond
 Macro referenced in 7a.

$\langle \text{implementation of Equation (5d) } 7e \rangle \equiv$
ld outlorentz(-1, iv?, k1?, m?) *
 out(field1?, k1?, -1) =
 NCContainer(USpa(k1, -1), iv);
ld inplorentz(-1, iv?, k1?, m?) *
 inp(field1?, k1?, -1) =
 NCContainer(UbarSpb(k1, -1), iv); \diamond
 Macro referenced in 7a.

4. SPIN-1 PARTICLES

For ingoing gauge bosons we use the polarisation vector $\varepsilon_\mu(k, j_3)$, and for outgoing particles its conjugate $\varepsilon_\mu^*(k, j_3)$ in accordance with the notation of [1]. For internal particles we work in Feynman gauge and hence get the propagator

$$(6) \quad \frac{-ig^{\mu\nu}}{k^2 - m^2 - im\Gamma + i0^+}.$$

$\langle \text{gauge boson propagator } 8a \rangle \equiv$
`ld proplorentz(2, k1?, m?, sDUMMY1?, 0, iv1?, iv2?) =`
 `- PREFACTOR(i_) * d(iv1, iv2) * inv(k1, m, sDUMMY1);`
`ld proplorentz(2, 0, m?, sDUMMY1?, 0, iv1?, iv2?) =`
 `- PREFACTOR(i_) * d(iv1, iv2) * inv(ZERO, m, sDUMMY1);`
 Macro referenced in 3.

4.1. Massless Case. We represent massless gauge bosons in the way proposed by [2],

$$(7) \quad \varepsilon_\mu(k, +1) = \frac{\langle q | \gamma_\mu | k \rangle}{\sqrt{2} \langle qk \rangle}$$

$$(8) \quad \varepsilon_\mu(k, -1) = \frac{[q | \gamma_\mu | k \rangle}{\sqrt{2} [kq]}$$

which requires an arbitrary, light-like auxilliary vector q . It follows that in this representation

$$(9) \quad (\varepsilon_\mu(k, \pm 1))^* = \varepsilon_\mu(k, \mp 1).$$

Below we implement the above expressions with the notation $k = \mathbf{k1}$, $q = \mathbf{vDUMMY1}$ and $\mu = \mathbf{ivL2}$.

$\langle \text{expression for } \varepsilon(k, +1) \text{ } 8b \rangle \equiv$
`1/sqrt2 * SpDenominator(Spa2(vDUMMY1, k1)) *`
 `UbarSpa(vDUMMY1) * Sm(ivL2) * USpb(k1)`
 Macro referenced in 9ab.

$\langle \text{expression for } \varepsilon(k, -1) \text{ } 8c \rangle \equiv$
`1/sqrt2 * SpDenominator(Spb2(k1, vDUMMY1)) *`
 `UbarSpb(vDUMMY1) * Sm(ivL2) * USpa(k1)`
 Macro referenced in 8d, 9b.

Using Equation (9) we can also define macros for the conjugate vectors

$\langle \text{expression for } \varepsilon^*(k, +1) \text{ } 8d \rangle \equiv$
 $\langle \text{expression for } \varepsilon(k, -1) \text{ } 8c \rangle \diamond$
 Macro referenced in 9b.

$\langle \text{expression for } \varepsilon^*(k, -1) \text{ 9a} \rangle \equiv$
 $\langle \text{expression for } \varepsilon(k, +1) \text{ 8b} \rangle \diamond$
 Macro referenced in 9b.

The usual properties for polarisation vectors are easy to prove. The polarisation vector is transverse both to k and q :

$$(10) \quad \varepsilon(k, j_3) \cdot k = 0,$$

$$(11) \quad \varepsilon(k, j_3) \cdot q = 0.$$

The polarisation vectors fulfill the completeness relation of an axial gauge,

$$(12) \quad \sum_{j_3=\pm 1} \varepsilon^\mu(k, j_3) (\varepsilon^\nu(k, j_3))^* = -g^{\mu\nu} + \frac{k^\mu q^\nu + k^\nu q^\mu}{k \cdot q}.$$

By making use of the Schouten identity one can show that a change of the auxilliary vector $q \rightarrow p$ amounts to a term proportional to k_μ ,

$$(13a) \quad \frac{\langle q | \gamma_\mu | k \rangle}{\sqrt{2} \langle qk \rangle} = \frac{\langle p | \gamma_\mu | k \rangle}{\sqrt{2} \langle pk \rangle} + \frac{\sqrt{2} \langle pq \rangle}{\langle pk \rangle \langle qk \rangle} k_\mu$$

$$(13b) \quad \frac{[q | \gamma_\mu | k]}{\sqrt{2} [kq]} = \frac{[p | \gamma_\mu | k]}{\sqrt{2} [kp]} + \frac{\sqrt{2} [qp]}{[kp] [kq]} k_\mu$$

$\langle \text{gauge boson wave-functions, light-like 9b} \rangle \equiv$
 $\text{ld outlorentz}(2, \text{ivL2?}, \text{k1?}, 0) *$
 $\quad \text{out}(\text{field1?}, \text{k1?}, 1, \text{vDUMMY1?}) =$
 $\quad \langle \text{expression for } \varepsilon^*(k, +1) \text{ 8d} \rangle;$
 $\text{ld outlorentz}(2, \text{ivL2?}, \text{k1?}, 0) *$
 $\quad \text{out}(\text{field1?}, \text{k1?}, -1, \text{vDUMMY1?}) =$
 $\quad \langle \text{expression for } \varepsilon^*(k, -1) \text{ 9a} \rangle;$
 $\text{ld inplorentz}(2, \text{ivL2?}, \text{k1?}, 0) *$
 $\quad \text{inp}(\text{field1?}, \text{k1?}, 1, \text{vDUMMY1?}) =$
 $\quad \langle \text{expression for } \varepsilon(k, +1) \text{ 8b} \rangle;$
 $\text{ld inplorentz}(2, \text{ivL2?}, \text{k1?}, 0) *$
 $\quad \text{inp}(\text{field1?}, \text{k1?}, -1, \text{vDUMMY1?}) =$
 $\quad \langle \text{expression for } \varepsilon(k, -1) \text{ 8c} \rangle; \diamond$

Macro referenced in 2b.

4.2. Massive Case. For the polarisation vectors of massive gauge bosons, where $k^2 = m^2$, we require²

$$\begin{aligned}
 (14) \quad & \varepsilon(k, j_3) \cdot k = 0 && \text{transversality,} \\
 (15) \quad & \varepsilon(k, j_3) \cdot \varepsilon(k, j'_3) = -\delta_{j_3 j'_3} && \text{orthonormality and} \\
 (16) \quad & \sum_{j_3=-1}^1 \varepsilon_\mu(k, j_3) (\varepsilon_\nu(k, j_3))^* = -g^{\mu\nu} + \frac{k^\mu k^\nu}{m^2} && \text{completeness.}
 \end{aligned}$$

We choose a representation based on the splitting of the momentum k into a pair of light-like vector k^b and q , as it is implemented in the spinney procedure `SpLightConeDecomposition`.

$$(17) \quad k = k^b + \frac{m^2}{2k^b \cdot q} q$$

Similarly to the massless case, two of the polarisations can be chosen as

$$(18) \quad \varepsilon_\mu(k, +1) = \frac{\langle q | \gamma_\mu | k^b \rangle}{\sqrt{2} \langle q k^b \rangle} \quad \text{and}$$

$$(19) \quad \varepsilon_\mu(k, -1) = \frac{[q | \gamma_\mu | k^b \rangle}{\sqrt{2} [k^b q]}.$$

As before, these vectors have the property that they are complex conjugate to each other. The third polarisation vector is

$$(20) \quad \varepsilon_\mu(k, 0) = \frac{1}{m} \left(k_\mu^b - \frac{m^2}{2k^b \cdot q} q_\mu \right) = \frac{1}{m} (2k_\nu^b - k_\nu).$$

In the implementation we have $\mu = \text{ivL2}$, $k = \text{k1}$, $k^b = \text{k2}$ and $q = \text{k3}$.

$\langle \text{expression for massive } \varepsilon(k, +1) \text{ 10a} \rangle \equiv$
 $(1/\text{sqrt2} * \text{SpDenominator}(\text{Spa2}(\text{k3}, \text{k2}))) *$
 $\text{UbarSpa}(\text{k3}) * \text{Sm}(\text{ivL2}) * \text{USpb}(\text{k2}) \diamond$
 Macro referenced in [11bd](#), [14](#), [15](#), [20a](#).

$\langle \text{expression for massive } \varepsilon(k, -1) \text{ 10b} \rangle \equiv$
 $(1/\text{sqrt2} * \text{SpDenominator}(\text{Spb2}(\text{k2}, \text{k3}))) *$
 $\text{UbarSpb}(\text{k3}) * \text{Sm}(\text{ivL2}) * \text{USpa}(\text{k2}) \diamond$
 Macro referenced in [11ad](#), [14](#), [15](#), [20a](#).

$\langle \text{expression for massive } \varepsilon(k, 0) \text{ 10c} \rangle \equiv$
 $(1/m) * (\text{k2}(\text{ivL2}) - m * \text{SpDenominator}(\text{Spa2}(\text{k2}, \text{k3}))) *$
 $m * \text{SpDenominator}(\text{Spb2}(\text{k3}, \text{k2})) * \text{k3}(\text{ivL2}) \diamond$
 Macro referenced in [11cd](#), [14](#), [15](#), [20a](#).

²See for example Appendix A.1.1.6 of [1].

The conjugate polarisation vectors are as follows.

$$\langle \text{expression for massive } \varepsilon^*(k, +1) \text{ 11a} \rangle \equiv \langle \text{expression for massive } \varepsilon(k, -1) \text{ 10b} \rangle \diamond$$

Macro referenced in 11d, 16, 17, 20c.

$$\langle \text{expression for massive } \varepsilon^*(k, -1) \text{ 11b} \rangle \equiv \langle \text{expression for massive } \varepsilon(k, +1) \text{ 10a} \rangle \diamond$$

Macro referenced in 11d, 16, 17, 20c.

$$\langle \text{expression for massive } \varepsilon^*(k, 0) \text{ 11c} \rangle \equiv \langle \text{expression for massive } \varepsilon(k, 0) \text{ 10c} \rangle \diamond$$

Macro referenced in 11d, 16, 17, 20c.

Finally, we can express all six possibilities of initial state and final state polarisation vectors:

$$\begin{aligned} \langle \text{gauge boson wave-functions, massive 11d} \rangle &\equiv \\ \text{ld outlorentz}(2, \text{ivL2?}, \text{k1?}, \text{m?}) * & \\ \text{out}(\text{field1?}, \text{k1?}, 1, \text{k2?}, \text{k3?}) = & \\ \langle \text{expression for massive } \varepsilon^*(k, +1) \text{ 11a} \rangle; & \\ \text{ld outlorentz}(2, \text{ivL2?}, \text{k1?}, \text{m?}) * & \\ \text{out}(\text{field1?}, \text{k1?}, -1, \text{k2?}, \text{k3?}) = & \\ \langle \text{expression for massive } \varepsilon^*(k, -1) \text{ 11b} \rangle; & \\ \text{ld outlorentz}(2, \text{ivL2?}, \text{k1?}, \text{m?}) * & \\ \text{out}(\text{field1?}, \text{k1?}, 0, \text{k2?}, \text{k3?}) = & \\ \langle \text{expression for massive } \varepsilon^*(k, 0) \text{ 11c} \rangle; & \\ \text{ld inplorentz}(2, \text{ivL2?}, \text{k1?}, \text{m?}) * & \\ \text{inp}(\text{field1?}, \text{k1?}, 1, \text{k2?}, \text{k3?}) = & \\ \langle \text{expression for massive } \varepsilon(k, +1) \text{ 10a} \rangle; & \\ \text{ld inplorentz}(2, \text{ivL2?}, \text{k1?}, \text{m?}) * & \\ \text{inp}(\text{field1?}, \text{k1?}, -1, \text{k2?}, \text{k3?}) = & \\ \langle \text{expression for massive } \varepsilon(k, -1) \text{ 10b} \rangle; & \\ \text{ld inplorentz}(2, \text{ivL2?}, \text{k1?}, \text{m?}) * & \\ \text{inp}(\text{field1?}, \text{k1?}, 0, \text{k2?}, \text{k3?}) = & \\ \langle \text{expression for massive } \varepsilon(k, 0) \text{ 10c} \rangle; \diamond & \end{aligned}$$

Macro referenced in 2b.

TODO:

*In cases where no massless vectors are in the process **GOLEM** chooses the procedure **SpLightConeSplitting** where a pair of massive vectors P, Q is split into a pair of light-like vectors p, q . The corresponding formulæ for polarisation vectors have to be worked out. Since this case is for very specific processes only we leave this for the future.*

5. SPIN- $\frac{3}{2}$ PARTICLES

For the implementation of massive Spin- $\frac{3}{2}$ fields we follow [3].

The projector is

$$(21) \quad \Pi^{\mu\nu} = (\not{p} + m) \left(\frac{p^\mu p^\nu}{m^2} - g^{\mu\nu} \right) - \frac{1}{3} \left(\gamma^\mu + \frac{k^\mu}{m} \right) (\not{k} - m) \left(\gamma^\nu + \frac{k^\nu}{m} \right).$$

$\langle \text{vector-spinor propagators 12} \rangle \equiv$

Repeat;

Id once proplorentz(3, k1?, m?, sDUMMY1?, 0, iv1?, iv2?) =

PREFACTOR(i_) *

SplitLorentzIndex(iv1, iv1L2, iv1L1) *

SplitLorentzIndex(iv2, iv2L2, iv2L1) * 1/3 * (

+ 4*k1(iv1L2)*k1(iv2L2)/csqrt(m*(m-i_*sDUMMY1))

- 3*d(iv1L2,iv2L2)*csqrt(m*(m-i_*sDUMMY1))

+ 2*NCCContainer(Sm(k1),iv1L1,iv2L1)*k1(iv1L2)*k1(iv2L2)/csqrt(m*(m-i_*sDUMMY1))

- 3*NCCContainer(Sm(k1),iv1L1,iv2L1)*d(iv1L2,iv2L2)

- NCCContainer(Sm(k1)*Sm(iv2L2),iv1L1,iv2L1)*k1(iv1L2)/csqrt(m*(m-i_*sDUMMY1))

+ NCCContainer(Sm(iv1L2),iv1L1,iv2L1)*k1(iv2L2)

- NCCContainer(Sm(iv1L2)*Sm(k1),iv1L1,iv2L1)*k1(iv2L2)/csqrt(m*(m-i_*sDUMMY1))

- NCCContainer(Sm(iv1L2)*Sm(k1)*Sm(iv2L2),iv1L1,iv2L1)

+ NCCContainer(Sm(iv1L2)*Sm(iv2L2),iv1L1,iv2L1)*csqrt(m*(m-i_*sDUMMY1))

+ NCCContainer(Sm(iv2L2),iv1L1,iv2L1)*k1(iv1L2)

) * inv(k1, m, sDUMMY1);

Sum iv1L2, iv1L1, iv2L2, iv2L1;

EndRepeat;

Repeat;

Id once proplorentz(3, k1?, m?, 0, 0, iv1?, iv2?) =

PREFACTOR(i_) *

SplitLorentzIndex(iv1, iv1L2, iv1L1) *

SplitLorentzIndex(iv2, iv2L2, iv2L1) * 1/3 * (

+ 4*k1(iv1L2)*k1(iv2L2)/m

- 3*d(iv1L2,iv2L2)*m

+ 2*NCCContainer(Sm(k1),iv1L1,iv2L1)*k1(iv1L2)*k1(iv2L2)/m^2

- 3*NCCContainer(Sm(k1),iv1L1,iv2L1)*d(iv1L2,iv2L2)

- NCCContainer(Sm(k1)*Sm(iv2L2),iv1L1,iv2L1)*k1(iv1L2)/m

+ NCCContainer(Sm(iv1L2),iv1L1,iv2L1)*k1(iv2L2)

- NCCContainer(Sm(iv1L2)*Sm(k1),iv1L1,iv2L1)*k1(iv2L2)/m

- NCCContainer(Sm(iv1L2)*Sm(k1)*Sm(iv2L2),iv1L1,iv2L1)

+ NCCContainer(Sm(iv1L2)*Sm(iv2L2),iv1L1,iv2L1)*m

+ NCCContainer(Sm(iv2L2),iv1L1,iv2L1)*k1(iv1L2)

) * inv(k1, m, 0);

Sum iv1L2, iv1L1, iv2L2, iv2L1;

EndRepeat;◇

Macro referenced in 3.

A set of eigenvectors is provided by the following five states:

$$(22) \quad \epsilon_{+3/2}^\mu(p) = \epsilon_+^\mu(p)\epsilon_+(p)$$

$$(23) \quad \epsilon_{+1/2}^\mu(p) = \frac{1}{\sqrt{3}}\epsilon_+^\mu(p)\epsilon_-(p) + \sqrt{\frac{2}{3}}\epsilon_0^\mu(p)\epsilon_+(p)$$

$$(24) \quad \epsilon_{-1/2}^\mu(p) = \frac{1}{\sqrt{3}}\epsilon_-^\mu(p)\epsilon_+(p) + \sqrt{\frac{2}{3}}\epsilon_0^\mu(p)\epsilon_-(p)$$

$$(25) \quad \epsilon_{-3/2}^\mu(p) = \epsilon_-^\mu(p)\epsilon_-(p)$$

There are sixteen different cases

- in-/outgoing
- particle/anti-particle
- polarisation $-\frac{3}{2}, -\frac{1}{2}, \frac{1}{2}, \frac{3}{2}$

$\langle \text{vector-spinor wave functions } 13 \rangle \equiv$
 $\langle \text{ingoing vector-spinor particle } 14 \rangle$
 $\langle \text{ingoing vector-spinor anti-particle } 15 \rangle$
 $\langle \text{outgoing vector-spinor particle } 16 \rangle$
 $\langle \text{outgoing vector-spinor anti-particle } 17 \rangle \diamond$

Macro referenced in [2b](#).

$\langle \text{ingoing vector-spinor particle } 14 \rangle \equiv$

```

Id once inplorentz(3, ivL?, k1?, m?!\{0,\}) *
    inp(field1?, k1?, -2, k2?, k3?) =
    SplitLorentzIndex(ivL, ivL2, ivL1) *
     $\langle \text{expression for massive } \varepsilon(k, -1) \text{ 10b} \rangle$  *
    NCContainer(USpb(k1,+1), ivL1);
Sum ivL2, ivL1;

Id once inplorentz(3, ivL?, k1?, m?!\{0,\}) *
    inp(field1?, k1?, -1, k2?, k3?) = 1/sqrt3 *
    SplitLorentzIndex(ivL, ivL2, ivL1) * (
    +  $\langle \text{expression for massive } \varepsilon(k, -1) \text{ 10b} \rangle$  *
    NCContainer(USpa(k1,+1), ivL1)
    + sqrt2 *  $\langle \text{expression for massive } \varepsilon(k, 0) \text{ 10c} \rangle$  *
    NCContainer(USpb(k1,+1), ivL1));
Sum ivL2, ivL1;

Id once inplorentz(3, ivL?, k1?, m?!\{0,\}) *
    inp(field1?, k1?, +1, k2?, k3?) = 1/sqrt3 *
    SplitLorentzIndex(ivL, ivL2, ivL1) * (
    +  $\langle \text{expression for massive } \varepsilon(k, +1) \text{ 10a} \rangle$  *
    NCContainer(USpb(k1,+1), ivL1)
    + sqrt2 *  $\langle \text{expression for massive } \varepsilon(k, 0) \text{ 10c} \rangle$  *
    NCContainer(USpa(k1,+1), ivL1));
Sum ivL2, ivL1;

Id once inplorentz(3, ivL?, k1?, m?!\{0,\}) *
    inp(field1?, k1?, +2, k2?, k3?) =
    SplitLorentzIndex(ivL, ivL2, ivL1) *
     $\langle \text{expression for massive } \varepsilon(k, +1) \text{ 10a} \rangle$  *
    NCContainer(USpa(k1,+1), ivL1);
Sum ivL2, ivL1;

```

◇

Macro referenced in 13.

$\langle \text{ingoing vector-spinor anti-particle } 15 \rangle \equiv$
Id once inplorentz(-3, ivL?, k1?, m?!{0,}) *
 inp(field1?, k1?, -2, k2?, k3?) =
 SplitLorentzIndex(ivL, ivL2, ivL1) *
 $\langle \text{expression for massive } \varepsilon(k, -1) \text{ 10b} \rangle$ *
 NCContainer(UbarSpb(k1, -1), ivL1);
 Sum ivL2, ivL1;
Id once inplorentz(-3, ivL?, k1?, m?!{0,}) *
 inp(field1?, k1?, -1, k2?, k3?) = 1/sqrt3 *
 SplitLorentzIndex(ivL, ivL2, ivL1) * (
 + $\langle \text{expression for massive } \varepsilon(k, -1) \text{ 10b} \rangle$ *
 NCContainer(UbarSpa(k1, -1), ivL1)
 + sqrt2 * $\langle \text{expression for massive } \varepsilon(k, 0) \text{ 10c} \rangle$ *
 NCContainer(UbarSpb(k1, -1), ivL1));
 Sum ivL2, ivL1;
Id once inplorentz(-3, ivL?, k1?, m?!{0,}) *
 inp(field1?, k1?, +1, k2?, k3?) = 1/sqrt3 *
 SplitLorentzIndex(ivL, ivL2, ivL1) * (
 + $\langle \text{expression for massive } \varepsilon(k, +1) \text{ 10a} \rangle$ *
 NCContainer(UbarSpb(k1, -1), ivL1)
 + sqrt2 * $\langle \text{expression for massive } \varepsilon(k, 0) \text{ 10c} \rangle$ *
 NCContainer(UbarSpa(k1, -1), ivL1));
 Sum ivL2, ivL1;
Id once inplorentz(-3, ivL?, k1?, m?!{0,}) *
 inp(field1?, k1?, +2, k2?, k3?) =
 SplitLorentzIndex(ivL, ivL2, ivL1) *
 $\langle \text{expression for massive } \varepsilon(k, +1) \text{ 10a} \rangle$ *
 NCContainer(UbarSpa(k1, -1), ivL1);
 Sum ivL2, ivL1;

◇

Macro referenced in 13.

$\langle \text{outgoing vector-spinor particle } 16 \rangle \equiv$
Id once inplorentz(3, ivL?, k1?, m?!\{0,\}) *
 inp(field1?, k1?, -2, k2?, k3?) =
 SplitLorentzIndex(ivL, ivL2, ivL1) *
 $\langle \text{expression for massive } \varepsilon^*(k, -1) \text{ 11b} \rangle$ *
 NCContainer(UbarSpa(k1,+1), ivL1);
 Sum ivL2, ivL1;
Id once inplorentz(3, ivL?, k1?, m?!\{0,\}) *
 inp(field1?, k1?, -1, k2?, k3?) = 1/sqrt3 *
 SplitLorentzIndex(ivL, ivL2, ivL1) * (
 + $\langle \text{expression for massive } \varepsilon^*(k, -1) \text{ 11b} \rangle$ *
 NCContainer(UbarSpb(k1,+1), ivL1)
 + sqrt2 * $\langle \text{expression for massive } \varepsilon^*(k, 0) \text{ 11c} \rangle$ *
 NCContainer(UbarSpa(k1,+1), ivL1));
 Sum ivL2, ivL1;
Id once inplorentz(3, ivL?, k1?, m?!\{0,\}) *
 inp(field1?, k1?, +1, k2?, k3?) = 1/sqrt3 *
 SplitLorentzIndex(ivL, ivL2, ivL1) * (
 + $\langle \text{expression for massive } \varepsilon^*(k, +1) \text{ 11a} \rangle$ *
 NCContainer(UbarSpa(k1,+1), ivL1)
 + sqrt2 * $\langle \text{expression for massive } \varepsilon^*(k, 0) \text{ 11c} \rangle$ *
 NCContainer(UbarSpb(k1,+1), ivL1));
 Sum ivL2, ivL1;
Id once inplorentz(3, ivL?, k1?, m?!\{0,\}) *
 inp(field1?, k1?, +2, k2?, k3?) =
 SplitLorentzIndex(ivL, ivL2, ivL1) *
 $\langle \text{expression for massive } \varepsilon^*(k, +1) \text{ 11a} \rangle$ *
 NCContainer(UbarSpb(k1,+1), ivL1);
 Sum ivL2, ivL1;

◇

Macro referenced in 13.

$\langle \text{outgoing vector-spinor anti-particle } 17 \rangle \equiv$

```

Id once inplorentz(3, ivL?, k1?, m?!{0,}) *
    inp(field1?, k1?, -2, k2?, k3?) =
    SplitLorentzIndex(ivL, ivL2, ivL1) *
     $\langle \text{expression for massive } \varepsilon^*(k, -1) \text{ 11b} \rangle$  *
    NCContainer(USpa(k1, -1), ivL1);
Sum ivL2, ivL1;

Id once inplorentz(3, ivL?, k1?, m?!{0,}) *
    inp(field1?, k1?, -1, k2?, k3?) = 1/sqrt3 *
    SplitLorentzIndex(ivL, ivL2, ivL1) * (
    +  $\langle \text{expression for massive } \varepsilon^*(k, -1) \text{ 11b} \rangle$  *
    NCContainer(USpb(k1, -1), ivL1)
    + sqrt2 *  $\langle \text{expression for massive } \varepsilon^*(k, 0) \text{ 11c} \rangle$  *
    NCContainer(USpa(k1, -1), ivL1));
Sum ivL2, ivL1;

Id once inplorentz(3, ivL?, k1?, m?!{0,}) *
    inp(field1?, k1?, +1, k2?, k3?) = 1/sqrt3 *
    SplitLorentzIndex(ivL, ivL2, ivL1) * (
    +  $\langle \text{expression for massive } \varepsilon^*(k, +1) \text{ 11a} \rangle$  *
    NCContainer(USpa(k1, -1), ivL1)
    + sqrt2 *  $\langle \text{expression for massive } \varepsilon^*(k, 0) \text{ 11c} \rangle$  *
    NCContainer(USpb(k1, -1), ivL1));
Sum ivL2, ivL1;

Id once inplorentz(3, ivL?, k1?, m?!{0,}) *
    inp(field1?, k1?, +2, k2?, k3?) =
    SplitLorentzIndex(ivL, ivL2, ivL1) *
     $\langle \text{expression for massive } \varepsilon^*(k, +1) \text{ 11a} \rangle$  *
    NCContainer(USpb(k1, -1), ivL1);
Sum ivL2, ivL1;

```

◇

Macro referenced in 13.

6. SPIN-2 PARTICLES

6.1. Tensor Structure. In order to map the pair of Lorentz indices into a single Multi-Index we use the function `SplitLorentzIndex`; the first argument denotes the multi-index, the second and the last argument are the two Lorentz indices.

6.2. Tensor Ghost. The CalcHEP way of treating colour requires the introduction of the so-called *tensor ghost*. This auxilliary field is introduced in order to split the four-gluon vertex into a pair of gluon-gluon-ghost vertices. The propagator therefore is not dynamical and has the form

$$(26) \quad P(T^{\mu_1 \nu_1}(p_1), T^{\mu_2 \nu_2}(p_2)) = -ig^{\mu_1 \mu_2} g^{\nu_1 \nu_2}$$

Tensor ghosts are indicated by having an auxilliary field value of 1.

$\langle \text{tensor ghost propagator 18a} \rangle \equiv$

```

Id once proplorentz(4, k1?, m?, sDUMMY1?, 1, iv1?, iv2?) =
  - PREFACTOR(i_) *
    SplitLorentzIndex(iv1, iv1a, iv1b) *
    SplitLorentzIndex(iv2, iv2a, iv2b) *
    d(iv1a, iv2a) * d(iv1b, iv2b);
Sum iv1a, iv1b, iv2a, iv2b;◇
Macro referenced in 3.

```

6.3. Gravitons.

$$\begin{aligned}
 \Pi^{\mu\nu,\alpha\beta}(p) &= \frac{1}{2} \left(g^{\mu\alpha} - \frac{p^\mu p^\alpha}{m^2} \right) \left(g^{\nu\beta} - \frac{p^\nu p^\beta}{m^2} \right) \\
 &+ \frac{1}{2} \left(g^{\mu\beta} - \frac{p^\mu p^\beta}{m^2} \right) \left(g^{\nu\alpha} - \frac{p^\nu p^\alpha}{m^2} \right) \\
 (27) \quad &- \frac{1}{3} \left(g^{\mu\nu} - \frac{p^\mu p^\nu}{m^2} \right) \left(g^{\alpha\beta} - \frac{p^\alpha p^\beta}{m^2} \right)
 \end{aligned}$$

$\langle \text{graviton propagator 18b} \rangle \equiv$

```

Id once proplorentz(4, k1?, m?, sDUMMY1?, 0, iv1?, iv2?) =
  SplitLorentzIndex(iv1, iv1a, iv1b) *
  SplitLorentzIndex(iv2, iv2a, iv2b) *
  (
    + 1/2 * (
      + (d_(iv1a, iv2a) - k1(iv1a)*k1(iv2a)/m^2) *
      (d_(iv1b, iv2b) - k1(iv1b)*k1(iv2b)/m^2) *
      + (d_(iv1a, iv2b) - k1(iv1a)*k1(iv2b)/m^2) *
      (d_(iv1b, iv2a) - k1(iv1b)*k1(iv2a)/m^2)
    )
    - 1/3 * (
      + (d_(iv1a, iv1b) - k1(iv1a)*k1(iv1b)/m^2) *
      (d_(iv2a, iv2b) - k1(iv2a)*k1(iv2b)/m^2)
    )
  ) * inv(k1, m, sDUMMY1);
◇
Macro referenced in 3.

```

$$(28) \quad \epsilon_{\pm 2}^{\mu\nu}(p) = \epsilon_{\pm}^{\mu}(p) \epsilon_{\pm}^{\nu}(p)$$

$$(29) \quad \epsilon_{\pm 1}^{\mu\nu}(p) = \frac{1}{\sqrt{2}} (\epsilon_{\pm}^{\mu}(p) \epsilon_0^{\nu}(p) + \epsilon_{\pm}^{\nu}(p) \epsilon_0^{\mu}(p))$$

$$(30) \quad \epsilon_0^{\mu\nu}(p) = \frac{1}{\sqrt{6}} (\epsilon_+^{\mu}(p) \epsilon_-^{\nu}(p) + \epsilon_-^{\mu}(p) \epsilon_+^{\nu}(p) + 2\epsilon_0^{\mu}(p) \epsilon_0^{\nu}(p))$$

There is the possibility, to have custom propagators for Spin-2 particles. This is especially needed for the sum over KK modes in the ADD and RS model. For this the `customSpin2Prop` Fortran function need to be defined by hand. The mass terms in the Lorentz structure are neglected.

$\langle \text{graviton effective propagator 19a} \rangle \equiv$

```

Id once proplorentz(4, k1?, m?, sDUMMY1?, 2, iv1?, iv2?) =
  SplitLorentzIndex(iv1, iv1a, iv1b) *
  SplitLorentzIndex(iv2, iv2a, iv2b) *
  (
    + 1/2 * (
      + (d_(iv1a, iv2a) ) *
        (d_(iv1b, iv2b) )
      + (d_(iv1a, iv2b) ) *
        (d_(iv1b, iv2a) )
    )
    - 1/3 * (
      + (d_(iv1a, iv1b) ) *
        (d_(iv2a, iv2b) )
    )
  ) * customSpin2Prop(k1, m, sDUMMY1);

```

◇

Macro referenced in [3](#).

We have to consider 10 cases:

- in-/outgoing
- polarisations $\pm 2, \pm 1, 0$

$\langle \text{graviton wave functions 19b} \rangle \equiv$

$\langle \text{ingoing graviton wave functions 20a} \rangle$

$\langle \text{outgoing graviton wave functions 20c} \rangle$ ◇

Macro referenced in [2b](#).

Since we have to distinguish two Lorentz indices we use auxilliary function `fDUMMY1` to denote $\epsilon^\mu(k, \lambda)$ and $\epsilon_\mu^*(k, \lambda)$ in the ingoing and outgoing case.

$\langle \text{generic graviton wave functions 19c} \rangle \equiv$

```

Id once @1lorentz(4, ivL4?, k1?, m?) *
  @1(field1?, k1?, +2, k2?, k3?) =
  SplitLorentzIndex(ivL4, ivL2a, ivL2b) *
  fDUMMY1(ivL2a, k1, +1, k2, k3, m) * fDUMMY1(ivL2b, k1, +1, k2, k3, m);
Sum ivL2a, ivL2b;
Id once @1lorentz(4, ivL4?, k1?, m?) *
  @1(field1?, k1?, +1, k2?, k3?) =
  SplitLorentzIndex(ivL4, ivL2a, ivL2b) *
  1/Sqrt2 * (

```

```

    + fDUMMY1(ivL2a, k1, +1, k2, k3, m) * fDUMMY1(ivL2b, k1, 0, k2, k3,
    + fDUMMY1(ivL2a, k1, 0, k2, k3, m) * fDUMMY1(ivL2b, k1, +1, k2, k3,
  );
  Sum ivL2, ivL1;
ld once @1lorentz(4, ivL4?, k1?, m?) *
    @1(field1?, k1?, 0, k2?, k3?) =
  SplitLorentzIndex(ivL4, ivL2a, ivL2b) *
  1/Sqrt2/Sqrt3 * (
    + fDUMMY1(ivL2a, k1, +1, k2, k3, m) * fDUMMY1(ivL2b, k1, -1, k2, k3,
    + fDUMMY1(ivL2a, k1, -1, k2, k3, m) * fDUMMY1(ivL2b, k1, +1, k2, k3,
    + 2 * fDUMMY1(ivL2a, k1, 0, k2, k3, m) * fDUMMY1(ivL2b, k1, 0, k2, k3,
  );
  Sum ivL2, ivL1;
ld once @1lorentz(4, ivL4?, k1?, m?) *
    @1(field1?, k1?, -1, k2?, k3?) =
  SplitLorentzIndex(ivL4, ivL2a, ivL2b) *
  1/Sqrt2 * (
    + fDUMMY1(ivL2a, k1, -1, k2, k3, m) * fDUMMY1(ivL2b, k1, 0, k2, k3,
    + fDUMMY1(ivL2a, k1, 0, k2, k3, m) * fDUMMY1(ivL2b, k1, -1, k2, k3,
  );
  Sum ivL2, ivL1;
ld once @1lorentz(4, ivL4?, k1?, m?) *
    @1(field1?, k1?, -2, k2?, k3?) =
  SplitLorentzIndex(ivL4, ivL2a, ivL2b) *
  fDUMMY1(ivL2a, k1, -1, k2, k3, m) * fDUMMY1(ivL2b, k1, -1, k2, k3, m);
  Sum ivL2a, ivL2b;
  ◇

```

Macro referenced in [20ac](#).

$\langle \text{ingoing graviton wave functions } 20a \rangle \equiv$
 $\langle \text{generic graviton wave functions } (20b \text{ inp}) \text{ } 19c \rangle$
ld fDUMMY1(ivL2?, k1?, +1, k2?, k3?, m?) =
 $\langle \text{expression for massive } \varepsilon(k, +1) \text{ } 10a \rangle$;
ld fDUMMY1(ivL2?, k1?, 0, k2?, k3?, m?) =
 $\langle \text{expression for massive } \varepsilon(k, 0) \text{ } 10c \rangle$;
ld fDUMMY1(ivL2?, k1?, -1, k2?, k3?, m?) =
 $\langle \text{expression for massive } \varepsilon(k, -1) \text{ } 10b \rangle$; ◇

Macro referenced in [19b](#).

$\langle \text{outgoing graviton wave functions } 20c \rangle \equiv$
 $\langle \text{generic graviton wave functions } (20d \text{ out}) \text{ } 19c \rangle$
 $\text{ld fDUMMY1}(ivL2?, k1?, +1, k2?, k3?, m?) =$
 $\langle \text{expression for massive } \varepsilon^*(k, +1) \text{ } 11a \rangle;$
 $\text{ld fDUMMY1}(ivL2?, k1?, 0, k2?, k3?, m?) =$
 $\langle \text{expression for massive } \varepsilon^*(k, 0) \text{ } 11c \rangle;$
 $\text{ld fDUMMY1}(ivL2?, k1?, -1, k2?, k3?, m?) =$
 $\langle \text{expression for massive } \varepsilon^*(k, -1) \text{ } 11b \rangle; \diamond$
 Macro referenced in 19b.

7. THE COLOUR PART OF THE PROPAGATORS

This section is not at the main theme of this document but for historical reasons these replacement rules are expected in the file `propagators.hh`. The colour part of a propagator for all non-trivial representations is replaced by a Kronecker- δ . The trivial representation is just ignored.

$\langle \text{colour part of the propagators } 21 \rangle \equiv$
 $\text{ld propcolor}(3, iv1?, iv2?) = \mathbf{d_}(iv1, iv2);$
 $\text{ld propcolor}(-3, iv1?, iv2?) = \mathbf{d_}(iv1, iv2);$
 $\text{ld propcolor}(8, iv1?, iv2?) = \mathbf{d_}(iv1, iv2);$
 $\text{ld propcolor}(1, iv1?, iv2?) = 1; \diamond$
 Macro referenced in 3.

APPENDIX A. INDEX OF SYMBOLS

APPENDIX B. INDEX OF MACROS

$\langle \text{colour part of the propagators } 21 \rangle$ Referenced in 3.
 $\langle \text{common header } 2a \rangle$ Referenced in 2b, 3.
 $\langle \text{expression for } \varepsilon(k, +1) \text{ } 8b \rangle$ Referenced in 9ab.
 $\langle \text{expression for } \varepsilon(k, -1) \text{ } 8c \rangle$ Referenced in 8d, 9b.
 $\langle \text{expression for } \varepsilon^*(k, +1) \text{ } 8d \rangle$ Referenced in 9b.
 $\langle \text{expression for } \varepsilon^*(k, -1) \text{ } 9a \rangle$ Referenced in 9b.
 $\langle \text{expression for massive } \varepsilon(k, +1) \text{ } 10a \rangle$ Referenced in 11bd, 14, 15, 20a.
 $\langle \text{expression for massive } \varepsilon(k, -1) \text{ } 10b \rangle$ Referenced in 11ad, 14, 15, 20a.
 $\langle \text{expression for massive } \varepsilon(k, 0) \text{ } 10c \rangle$ Referenced in 11cd, 14, 15, 20a.
 $\langle \text{expression for massive } \varepsilon^*(k, +1) \text{ } 11a \rangle$ Referenced in 11d, 16, 17, 20c.
 $\langle \text{expression for massive } \varepsilon^*(k, -1) \text{ } 11b \rangle$ Referenced in 11d, 16, 17, 20c.
 $\langle \text{expression for massive } \varepsilon^*(k, 0) \text{ } 11c \rangle$ Referenced in 11d, 16, 17, 20c.
 $\langle \text{fermion propagator } 5a \rangle$ Referenced in 3.
 $\langle \text{gauge boson propagator } 8a \rangle$ Referenced in 3.
 $\langle \text{gauge boson wave-functions, light-like } 9b \rangle$ Referenced in 2b.
 $\langle \text{gauge boson wave-functions, massive } 11d \rangle$ Referenced in 2b.
 $\langle \text{generic graviton wave functions } 19c \rangle$ Referenced in 20ac.
 $\langle \text{graviton effective propagator } 19a \rangle$ Referenced in 3.
 $\langle \text{graviton propagator } 18b \rangle$ Referenced in 3.
 $\langle \text{graviton wave functions } 19b \rangle$ Referenced in 2b.

⟨ handed fermion propagator 5b ⟩ Referenced in 3.
 ⟨ implementation of Equation (4a) 6a ⟩ Referenced in 5c.
 ⟨ implementation of Equation (4b) 6b ⟩ Referenced in 5c.
 ⟨ implementation of Equation (4c) 6c ⟩ Referenced in 5c.
 ⟨ implementation of Equation (4d) 6d ⟩ Referenced in 5c.
 ⟨ implementation of Equation (5a) 7b ⟩ Referenced in 7a.
 ⟨ implementation of Equation (5b) 7c ⟩ Referenced in 7a.
 ⟨ implementation of Equation (5c) 7d ⟩ Referenced in 7a.
 ⟨ implementation of Equation (5d) 7e ⟩ Referenced in 7a.
 ⟨ ingoing graviton wave functions 20a ⟩ Referenced in 19b.
 ⟨ ingoing vector-spinor anti-particle 15 ⟩ Referenced in 13.
 ⟨ ingoing vector-spinor particle 14 ⟩ Referenced in 13.
 ⟨ outgoing graviton wave functions 20c ⟩ Referenced in 19b.
 ⟨ outgoing vector-spinor anti-particle 17 ⟩ Referenced in 13.
 ⟨ outgoing vector-spinor particle 16 ⟩ Referenced in 13.
 ⟨ scalar propagator 4b ⟩ Referenced in 3.
 ⟨ scalar wave-functions 4a ⟩ Referenced in 2b.
 ⟨ tensor ghost propagator 18a ⟩ Referenced in 3.
 ⟨ vector-spinor propagators 12 ⟩ Referenced in 3.
 ⟨ vector-spinor wave functions 13 ⟩ Referenced in 2b.
 ⟨ wave-functions for massive spinors 7a ⟩ Referenced in 2b.
 ⟨ wave-functions for massless spinors 5c ⟩ Referenced in 2b.

APPENDIX C. INDEX OF FILES

"legs.hh" Defined by 2b.
 "propagators.hh" Defined by 3.

REFERENCES

- [1] M. Böhm, A. Denner, H. Joos: *GAUGE THEORIES of the Strong and Electroweak Interaction*, B. G. Teubner, Stuttgart, Leipzig, Wiesbaden, 3rd edition, 2001.
- [2] Z. Xu, D. H. Zhang and L. Chang, *Helicity Amplitudes for Multiple Bremsstrahlung in Massless Nonabelian Gauge Theories*, Nucl. Phys. B **291** (1987) 392.
- [3] W. Kilian, T. Ohl and J. Reuter, *WHIZARD: Simulating Multi-Particle Processes at LHC and ILC*, arXiv:0708.4233 [hep-ph].
E-mail address: thomasr@nikhef.nl
URL: <http://www.nikhef.nl/~thomasr>